1st Teaching Workshop on Environmental Economics

for the Middle East and North Africa

December 5-16, 2005 - ICTP, Trieste, Italy

Valuation of coastal habitats sustaining plaice fisheries

Sandra Lerda

The Beijer International Institute of Ecological Economics

SMR 1684 - 10

Valuation of coastal habitats sustaining plaice fisheries

Valuation of coastal habitats sustaining plaice fisheries

Sandra Lerda

Introduction

The majority of the world's population lives in coastal areas and in the last few decades these areas have been seriously affected by all kinds of human activities, such as destruction of wetlands for agriculture, tourism and recreation, and waste disposal. The impact of human activities in coastal zones is causing habitat deterioration, affecting spawning grounds, nurseries and feeding grounds of marine resources, representing an increasing threat to global food security. (UNEP, Agenda 21: Protection of Oceans (Ch. 17))

The nature of the problems related to coastal zone management and the deterioration of our fish resource basis are subject to public policies, and often measures have to be taken in agreement among several countries. For this decision making process, it is useful to know the connections between human activities, habitat deterioration, and effects on social welfare. This paper deals with economic valuation of habitat changes that affect fish populations and ultimately social welfare, by analyzing the case of plaice in the West Coast of Sweden.

From an economic point of view, plaice (*Pleuronectes platessa*) is the most important flatfish in the areas of Kattegat and Skagerrak (between Sweden and Denmark). It is also a fish whose reproduction is seriously dependent on the habitat available for young individuals to grow and become part of the biomass swimming in the ocean. Juvenile plaice have spatially restricted nursery grounds located in shallow soft bottom areas. According to Wennhage et. al.[1], Swedish nursery grounds contribute with 77% to plaice recruitment in the area of Kattegat and Skagerrak.

---

1 Wennhage, Håkan, Leif Pihl and Johan Stål, "Distribution and quality of plaice (Pleuronectes platessa) nursery grounds on the Swedish west coast", manuscript, 2004.

In the west coast of Sweden seasonal algal blooms cover some of the areas where plaice larvae settle[2]. Since the settlement of plaice larvae in soft bottom areas is crucial for recruitment and population increase, increasing algae coverage implies a decrease in habitat quality, affecting the stock of plaice available at sea and consequently the fisheries.

In this paper I present a model connecting habitat quality, plaice population growth, and the plaice fishing activity. The model is written in GAMS and intends to be an instrument for valuation of an environmental change (presence of algal blooms) when the environment (soft bottom areas) is an input in the production process (plaice production). The most interesting aspect of this case study is the possibility to link - quantitatively and empirically - the ecological and the economic aspects of the problem in a bioeconomic model.

How are the plaice fisheries affected when algae coverage of the settlement areas increases, what is the economic and social impact of the habitat deterioration in the Swedish West Coast, what lessons can we take for public policy and, ultimately, how does ecosystem degradation affect society's wellbeing, are some of the questions this paper deals with.

Plaice: some biological and ecological aspects

Plaice (*Pleuronectes platessa*) is a fish whose reproduction is seriously dependent on the habitat available for young individuals to grow and become part of the biomass swimming in the ocean. Juvenile plaice have spatially restricted nursery grounds located in shallow soft bottom areas. As mentioned before, according to Wennhage et. al. (2004), Swedish nursery grounds contribute with 77% to plaice recruitment in the area of Kattegat and Skagerrak. In some of the bays in the west coast of Sweden, seasonal algal blooms cover these areas, affecting fish settlements. Preliminary studies show that if the coastal areas not covered by

[2] See, for example, Troell et. al., 2005.

algae are reduced[3], plaice recruitment would decrease and that would affect plaice production and the market equilibrium.

One way to value the contribution of the ecosystem to plaice production is to calculate the change in producer and consumer surplus produced by a change in the habitat availability or habitat quality, such as shown in Ellis and Fisher, 1987, Lynne et al. 1981, and Freeman, 1991. What I attempt in this paper is to build a bioeconomic dynamic model to serve as a basis for this valuation exercise.

As for the plaice population dynamics, we know from Pihl et. al (2005), that plaice larvae settle during the spring on the nursery grounds in the Swedish west coast and stay in these shallow areas until the autumn, when they move into deeper waters for spending the winter and continuing their life cycle. In the deeper sea, the "natural" mortality rate that plaice population is exposed to in its first two years of life is estimated to lie between 0.2 and 0.3. (personal communication with Håkan Wennhage). After 2 years, these 0-group juvenile fish are recruited to the adult population.

Another important aspect of plaice population dynamics is that there are no clear stock-recruitment relationships. As for other fish species, the size of the stock is determinant of the growth of the population. In what concerns plaice, though, only at very low stock levels is the stock a constraining variable explaining population growth. At "normal" stock levels, the habitat, the quantity and quality of shallow soft bottom areas available, is the bottleneck. The important aspect here is the density dependent growth and mortality in the nursery grounds (Pihl et. al. 2000). If the habitat available decreases, the density is higher, leading to higher mortality due to predation and lower growth due to food limitation.

The stock of plaice available at sea is kept at a relatively low level by the fishing pressure, meaning that factors like carrying capacity are not constraining population growth. Danish fishermen dominate the plaice fishery in the Swedish

[3] Wennhage, et al., 2004.

West Coast. Danish landings account for more than 90% of the total and the catch of plaice is carried out by three vessel categories: seiners, otter-trawlers and gillnetters. Plaice stock is assumed to be mature at the age of three years.

As mentioned before, the ecological problem that concerns us in this paper refers to the fact that, in some of the bays in the west coast of Sweden, seasonal algal blooms cover the areas where plaice larvae settles. Since the settlement of plaice larvae in soft bottom areas is crucial for recruitment and population increase, increasing algae coverage implies a decrease in habitat quality, affecting the stock of plaice available at sea and, consequently, the fisheries.

The effect of algae on plaice recruitment has been modeled by Pihl et. al. (2005). They have found that the relationship between vegetation coverage and recruitment of juvenile plaice depends on the behavior of the larvae in the presence of algae. Pihl et. al. model takes into consideration two different options for the behavior of the young fish in the settlements. In the first option ("stay"), the plaice that settles in vegetated areas stay there and perish, while the others settlers in non-algae areas are exposed to the "normal" mortality rate. The second option ("move") considers the possibility of fish moving from vegetated into non-vegetated areas. Instead of having its initial number reduced in direct relation to the algal mats, in this second option the fish would be exposed to a higher density dependent mortality.

Here I present only one of the options (the "move" option), for the sake of building the model, but one could change the equation for the vegetation-recruitment relationship to study the effect of the two options.

According to Pihl et al., the recruitment of 0-group juvenile plaice No_t is a function of the density dependent and density independent mortality rates (M), the size of the nursery area (A), the settlement density(D) and the vegetation cover (V):

$$No_t = A * D(1-V) * e^{(-M*t)} \qquad (1)$$

with mortality equal to

$$\mathcal{M} = 0.008 + 0.008 * \mathcal{L} \qquad (2)$$

Recruitment of two-year-old plaice would then be equal to:[4]

$$\mathcal{N}_t = \mathcal{N}o_t e^{(-2\mu)} = \mathcal{A} * \mathcal{D}(1 - \mathcal{V}) * e^{(-\mathcal{M}*t)} * e^{(-2\mu)} \qquad (3)$$

Where μ is a "natural" mortality rate that plaice population is exposed to in its first two years of life. It is estimated to lie between 0.2 and 0.3. (personal communication with Håkan Wennhage).

We leave for now the biological aspects of plaice population to discuss the fisheries model that serves at the basis for the exercise presented in this paper. These aspects will be used later for building different scenarios for how eutrophication affects the fishing economic activity.

The Beverton-Holt fisheries model

The most common model used in fisheries is the Schaefer model[5], based on the logistic growth curve which postulates an average relationship between the growth and the size of fish population, abstracting from its age structure.

The model used in this paper is based on Clark (1990, chapter 9), who presents what is known in the literature as the Beverton-Holt fisheries model. The Beverton-Holt model appeared first in 1957 and the authors' work laid the grounds for the modern age structured approach to optimal fisheries management[6].

---

[4] Here I still have a notation problem with the time between settlement and end of the summer, and time between the moment when these 0-year old plaice go deeper in the ocean and the moment they enter the stock (after two years)

[5] Schaefer's paper from 1954 is probably one of the most cited articles in fisheries economics. See for example Quinn, 2003; Knowler, 2002; and Eggert, 1998.

[6] See Beverton and Holt (1993).

The Beverton-Holt model describes the fish population as different cohorts for each year, resulting from the annual recruitment (R , the new additions to the fish stock). To follow the life-span of one cohort we have the number of fish in the cohort ($N(t)$) varying over time ($t \geq 0$) according to both the natural mortality rate ($M > 0$) and the fishing mortality rate ($F \geq 0$), in the following way:

$$\frac{dN}{dt} = -(M + F)N \qquad (4)$$

$$N(0) = R \qquad (5)$$

Both natural and fishing mortality are assumed to be independent of one another and in the simplest version presented by Clark, recruitment R is given.

The total biomass of each cohort is given by the total number of fish in a cohort multiplied by the average weight of a fish at age t. Clark uses the well-known von Bertalanffy weight function:

$$w(t) = a(1 - be^{-ct})^3 \qquad (6)$$

Where a, b, and c are positive constants.

The total biomass of a cohort ($B(t)$) in then given by $B(t) = N(t)w(t)$.

Clark shows the optimization over time of the harvest in such a simple one cohort fisheries model. The state equation for the dynamic problem is equation (4), slightly modified for allowing fishing mortality to be the control variable (harvest at different points in time):

$$\frac{dN}{dt} = -(M + F(t))N \qquad (7)$$

The objective function is the present value of the stream of profits that a fisherman can get out of fishing:

$$PV = \int_0^\infty e^{-\partial t}\left[pN(t)w(t) - C\right]F(t)dt$$

Where p is the constant price of fish, $w(t)$ is the average weight of a fish of age t and C is a constant cost coefficient.

Clark shows that the dynamic analysis of the multiple cohort Beverton-Holt model is much more complicated than the dynamics of the typical Schaefer fisheries model[7], even when using a set of simplifying assumptions (like for example that the cost of fishing is a constant coefficient). Searching for an analytical general result is a hard task. For that reason, in this paper I use a discrete dynamic model simulated in GAMS (General Algebraic Modelling System) for optimizing plaice harvest taking into consideration the population dynamics.

The GAMS model

Our problem is to find out what is the contribution of shallow soft bottom areas in the West Coast of Sweden to plaice fishery. In order to answer to this question, we must first establish the optimal harvest of plaice through time that maximizes the profits from the fishery. Assuming that the fishery is at this optimum, we then analyze different scenarios for habitat quality and the effect on the benefits from the fishery of changes in habitat.

For this purpose I build a discrete model to maximize the benefits from the fishing industry, subject to the fish population dynamics. Making $H(t)$ the harvest (kg) at time t, $C(t)$ the average cost of harvest at time t and p the constant per kg price of fish, we have $NB(t) = \sum_{t=2000}^{2034} \frac{1}{(1+r)^{t-2000}}(p.H(t) - C(t))$, the flow of net benefits from the fishery, as the objective function to be maximized.

[7] See for example page 292.

The price of plaice is assumed to be constant over time and internationally determined. The price used in the simulation is equal to SEK 13/kg, which corresponds roughly to the average price for plaice between the years 1996 and 2001, according to data from 2002 from the Danish Food and Resource Economic Institute.[8]

The time horizon used is 55 years, from 2000 to 2054, and t is the current time. k is the cohort time, the year the fish was born. I include in the model fish born since 1970, so as $1970 \leq k \leq$ 2054. $t-k$ is then the age of the fish. An average plaice is subject to a natural mortality and to the fishing mortality. I assume in the model that the natural mortality is equal to 0.1 and that every fish lives at the most for 30 years.

The discount rate in use is 3%, but this is something to be discussed later, since we use different discount rates to test the sensitivity of the model with respect to this parameter.

The fishing industry decision variable is harvest $H(t)$ and price is assumed to be nominally constant and determined by the international market. . The fish industry decides how much to harvest in order to maximize profits, subject to the restriction about the fish stock, determined by fish population dynamics.

The cost function used in the model, $C(t) = e^{-3,334} * H(t)^{0,9898} * TotB(t)^{-0,146}$, is the result of an estimation made for the period 1996-2001, with data about the Danish commercial plaice fisheries, compiled by the Danish Food and Resource Economic Institute.[9]

---

[8] Fødevareøkonomisk institut. For more information, please check "Economics analysis of fisheries in Kattegat and Skagerrak", FISHCASE project report, mimeo.

[9] For more detailed information about the estimation, please check "Economics analysis of fisheries in Kattegat and Skagerrak", FISHCASE project report, mimeo.

The restriction is the dynamics of the fish population: what is fished today affects what can be fished tomorrow, because it reduces the stock of fish available for the following periods, and so on. In the beginning of the time horizon the quantity harvested is big because the assumption used is that there was no harvest previous to the year 2000. The consequence of this is that the model assumes that there is a whole stock of fish accumulated to be harvested. After the year 2050, close to the end of the time horizon in the model, there is also no reason to preserve the fish stock anymore, and that is the reason why the whole remaining stock is fished in the last year.

$Q(k,t)$ corresponds to the number of fish from cohort k harvested at time t. It is important to keep in mind that the model assumes total selectivity of fishing gear, which means that I am assuming the very unlikely situation where the fisherman can choose from which cohort is the fish he is fishing. $H(t)=\sum_{k=1949}^{t} Q(k,t)*W(k,t)$ is the total harvest at time t, which is the sum of all harvested biomass from all cohorts at time t.

$N(k,t)=N(k,t-1)-Q(k,t-1)$ is the number of fish in a cohort, when t>k, that means, when that specific cohort is alive. Of course, when k>t, the number of fish in the cohort is $N(k,t)=0$.

The fish population increases through the yearly recruitment Rk. The number of fish in a given cohort the year that cohort is born (k=t) is the recruitment that year: $N(k,t)=Rk$. According to Wennhage et al.[10], yearly recruitment of 0-group plaice from Skagerrak and Kattegat nursery grounds is arouund 110 million individuals. Assuming a natural mortality rate between 0.2 and 0.3, this represents between 54 and 70 million 2 year old individuals entering the

[10] "Distribution and quality of plaice (*Pleuronectes platessa*) nursery grounds on the Swedish west coast, manuscript, 2005, p. 22.

population every year. Those are the numbers in use in the model as recruitment at time k.

Fish stock grows in numbers (through recruitment), but it also grows in weight, according to the von Bertalanffy weight function $W(k,t) = a(1 - be^{-c(t-k)})^3$. This function, which gives the average weight of a fish at age t-k, is bounded and increasing, with the proportional rate of increase in weight decreasing over time, as is shown in Figure 1. The maximum weight (in kg) of a plaice individual when times goes to infinite is given by the constant a, which I am assuming equal to 2.867.

Figure 1



$B(k,t) = N(k,t) * W(k,t)$ gives the fish biomass for the kth cohort, while the total fish biomass is given by $TotB(t) = \sum_{k}^{t} B(k,t)$.

Recruitment is assumed to be between 54 and 70 million individuals per year.[11]

Other assumptions being used in this model is that an individual plaice lives 30 years if it does not die from fishing or from natural mortality. The natural mortality parameter in use is equal to 0.1, which means that every year 10% of the fish population disappears from natural causes (diseases, predation, etc.). But this parameter can be changed and later in this paper I will explain why I am not using the natural mortality of 0.2 to 0.3 that was suggested as closer to reality.

Results

With the assumptions adopted and the built-in restrictions of the model, the plaice fishing industry in the west coast of Sweden would accumulate a total of 28261 millions of SEK of profits over the time horizon between 2000 and 2054 if it would harvest the resource following the optimal path suggested. The simulations with GAMS indicate, as expected, that the greater the fish recruitment, the greater the present value of the stream of profits from the fish industry over the time horizon.

As the following table shows, one more unit of fish recruitment (recall that the unit here is millions of individuals) implies SEK 524 million more of profits for the industry over the time horizon. That could be interpreted as the shadow price of plaice recruitment in the west coast of Sweden.

If we now link this result with the information we have about how recruitment is affected when there are algae blooms covering the nursery grounds and affecting habitat quality, we could easily get an indirect “price” for a square km of “clean”

[11] See Wennhage et al., submitted paper, p.22.

nursery ground. This would be the accounting price for one more square kilometer of soft bottom areas not affected by algal blooms.

Table: The shadow price of plaice recruitment

Value of Rk (In millions of individuals)	Value of the objective function (in millions of SEK)
54	28261.23
55	28785.74
56	29310.25
57	29834.76
58	30359.28
59	30883.79
60	31408.31
61	31932.82
62	32457.34
63	32981.86
64	33506.38
65	34030.90
66	34555.43
67	35079.95
68	35604.48
69	36129.00
70	36653.52

According to what was estimated by Wennhage et. al. (2005), the area of plaice nursery grounds along the west coast of Sweden is 157 km^2. Pihl et. al. (2005) investigated this area during three years and found that between 30% and 50% of this total area of potential nursery grounds in the west coast were covered by mats of filamentous algae. They also found that the reduction in the recruitment of juvenile plaice from nurseries can reach between 30% and 40%.

In a very rough calculation, using the results of the model, the presence of algal mats in the Swedish west coast could “cost” from 30% up to 40% of the total profits of the plaice fishing industry.

Discussion

This paper presented preliminary results of a simulation exercise about how one can value environmental changes in terms of economic production.

It is important to emphasize that these preliminary results are not only dependent on the restrictive assumptions already presented and used in the model. The conclusions in this paper are also dependent on other important assumptions that I would like to recall here. It is assumed that the current ecological conditions will remain the same in the future. It is also assumed that the economic conditions for the fish industry, including prices, are also the same in the future, as well as the institutional setting under which the economic agents take their decisions. The results for the model depend, therefore, on what kind of institutional set prevails: national or individual fishing quotas, maximum allowable catch, mesh restrictions, as well as other restrictions affecting fishing effort and so on.

References

Beverton, Raymond J. H., and Holt, Sidney J., *On the Dynamics of Exploited Fish Populations*, Chapman & Hall, 1993.

Clark, Colin W., *Mathematical Bioeconomics: the optimal management of renewable resources*, 2nd ed., John Wiley & Sons, 1990.

Eggert, Håkan, "Bioeconomic Analysis and Management: The case of Fisheries", *Environmental and Resource Economics*, 11(3-4):399-411, 1998.

Ellis, Gregory M. and Anthony C. Fisher, "Valuing the Environment as Input", *Journal of Environmental Management* 25(1987):149-156.

Freeman, A. Myrick III, "Valuing environmental resources under alternative management regimes", *Ecological Economics*, 3(1991):247-256.

Jensen, Carsten Lynge, "Applications of Dual Theory in Fisheries: A Survey", *Marine Resource Economics* 17:309-334.

Knowler, Duncan J, et. al., "Valuing freshwater salmon habitat on the west coast of Canada", *Journal of Environmental Management* 69 (2003): 261-273.

Knowler, Duncan, "A Review of Selected Bioeconomic Models with Environmental Influences in Fisheries", *Journal of Bioeconomics* 4 (2002):163-181

Lynne, Gary D. Et. Al., "Economic Valuation of March Areas for Marine Production Processes", *Journal of Environmental Economics and Management* 8 (1981):175-186.

Pihl, Leif, Johan Modin and Håkan Wennhage, "Relating plaice (Pleuronectes platessa) recruitment to deteriorating habitat quality: effects of macroalgal blooms in coastal nursery grounds, *Can. J. Fish. Aquat. Sci.* 62:1184-1193, 2005.

Quinn II, Terence J., "Ruminations on the development and future of population dynamics models in fisheries", *Natural Resource Modeling*, Vol. 16, No. 4 (winter 2003): 341-392.

Schaefer, M.B, "Some aspects of the dynamics of populations important to the management of commercial marine fisheries", *Bull. Inter-Amer. Trop. Tunna Comm.* 1:27–56, 1954.

Troell, Max, et. al., "Regime shifts and ecosystem services in Swedish coastal soft bottom habitats: when resilience is undesirable", *Ecology and Society* 10(1):30 [online] URL: http://www.ecologyandsociety.org/vol10/iss1/art30/

Wennhage, Håkan, Leif Pihl and Johan Stål, "Distribution and quality of plaice (Pleuronectes platessa) nursery grounds on the Swedish west coast", manuscript, 2005.

Appendix - GAMS code

Optimal Fish Harvesting subject to the von Bertalanffy weight function

```
$onupper

SETS    k       Time in which the kth cohort enters the fishery /1970*2054/
        t(k)    Time period of the model  /2000*2054/,
        tf(t)   First period of the model,
        tl(t)   Last period of the model,
        a(k,t)  Tuple indicating which cohorts are alive at time t;

*       Set up logic for the first and last period:

tf(t) = yes$(ord(t) = 1);
tl(t) = yes$(ord(t) = card(t));

SCALARS         p               Price of plaice         /13/
            r               Discount rate           /0.03/
            Rk              Recruitment at time k (millions)  /54/
            m               Natural fish mortality          /0.1/;

PARAMETER       yr(k)           Year corresponding to cohort k,
            w(k,t)          Weight of a cohort k fish in year t,
            pv(t)           Present value price;

yr(k) = ord(k)-1;
pv(t) = 1/(1+r)**ord(t);
w(k,t)$(yr(t) ge yr(k)) = 2.867 * (1 - EXP(-0.095*(yr(t)-yr(k)+1)))**3;

VARIABLES
        OBJ     Objective function;

POSITIVE VARIABLES
        N(k,t)  Number of fish belonging to the kth cohort at time t,
        B(k,t)  Fish biomass of the kth cohort,
        C(t)    Cost of fishing,
        Q(k,t)  Harvest in numbers,
        H(k,t)  Harvest biomass of fish from cohort k;

EQUATIONS
        EQNB            Defines the objective function
```

```
     EQN(k,t)        Equation for number of fish belonging to the kth cohort at time
t,
     EQB(k,t)        Equation for the fish biomass of the kth cohort,
     EQH(k,t)        Harvest biomass for a given cohort,
     EQC(t)          Equation for the cost function,
     QLAST(k,t)      Limit on harvest in final period;

EQNB ..              OBJ =E= SUM(t, pv(t) * (p * sum(a(k,t),H(k,t)) - C(t)));

EQN(k,t+1)$a(k,t)..     N(k,t+1) =E= N(k,t) - Q(k,t) - m * N(k,t);

QLAST(k,tl)$a(k,tl)..   N(k,tl) =g= Q(k,tl);

EQB(k,t)$a(k,t) ..      B(k,t)   =E= N(k,t) * w(k,t);

EQH(k,t)$a(k,t) ..      H(k,t)   =E= Q(k,t) * w(k,t);

*      Cost function

EQC(t)..        C(t) =E= SUM(a(k,t), exp(-3.334)* H(k,t))**(0.9898) *
                                    SUM(a(k,t), B(k,t))**(-0.146);

*        Introduce fish into the active population only when their biomass is
nonneglible:

a(k,t) = yes$(yr(k) le yr(t));

N.L(k,t)$a(k,t) = RK;

*       Assuming that fish cannot live more than 30 years:

B.L(k,t) = 0$(yr(t)- yr(k) ge 30);

*      Assume that no harvesting has occured prior to the first year:

N.FX(k,tf)$a(k,tf) = RK;
N.FX(t,t) = RK;

B.L(k,t)$a(k,t) = N.L(k,t) * w(k,t);
H.L(k,t)$a(k,t) = RK/10;

*      Avoid divide by zero errors:

H.LO(k,t) = 0.0001;
```

```
B.LO(k,t) = 0.0001;

MODEL PLAICE/ALL/;
SOLVE PLAICE USING DNLP MAXIMIZING OBJ;
DISPLAY OBJ.L;

$if not exist "%gams.sysdir%wgnupl32.exe" $exit

*       Produce some graphical output using GNUPLOT:

PARAMETER       wvalue(t,k)     Weight of Selected Cohorts Through the Model
Horizon;

set             kplot(k)                        Cohorts to be plotted
/1970,1980,1990,2000,2010,2020,2030,2040,2050/;

wvalue(t,kplot) = w(kplot,t);

set     tlbl(t) Time periods to label in plots /2000,2010,2020,2030,2040,2050/;

$setglobal gp_opt0 'set key outside width 3'
$setglobal domain t
$setglobal labels tlbl
$libinclude plot wvalue


PARAMETER       QH(t,k)         Fish Harvest Quantity,
                NF(t,k)         Numbers of Fish
                BF(t,k)         Biomass of Fish
                HF(t,k)         Harvest biomass,
                SUMMARY(t,*)    Summary statistics;

SUMMARY(t,"Q") = sum(k, Q.l(k,t));
SUMMARY(t,"N") = sum(k, N.l(k,t));
SUMMARY(t,"B") = sum(k, B.l(k,t));
SUMMARY(t,"H") = sum(k, H.l(k,t));
$libinclude plot summary

set   ksol(k)   Cohort solutions to plot

/1970,1975,1980,1985,1990,1995,2000,2005,2010,2015,2020,2025,2030,2040,20
50/;

QH(t,ksol) = na;
```

```
QH(t,ksol)$a(ksol,t) = Q.L(ksol,t);
$libinclude plot qh

NF(t,ksol) = na;
NF(t,ksol)$a(ksol,t) = N.L(ksol,t);
$libinclude plot nf

BF(t,ksol) = na;
BF(t,ksol)$a(ksol,t) = B.L(ksol,t);
$libinclude plot BF

HF(t,ksol) = na;
HF(t,ksol)$a(ksol,t) = H.L(ksol,t);
$libinclude plot HF
```